
1 INTRODUCTION 

The web service client of the 3-Heights™Document Converter Webservice provides document conversion services to 

SOAP based client applications. This tutorial shows the steps needed to integrate the 3-Heights™Document Converter 

Webservice into the SAP ABAP environment. 

 

2 PREREQUISITES  

The 3-Heights™Document Converter Webservice is accessible by the SAP Application Server on http resp. https port 

(check possible firewall constraints).  

  



joinApps,  08.04.2019 

     

 3-Heights™Document Converter Webservice Client Integration into SAP ABAP  

 

 

 

 

 
3 von 10 

  

 

 

3 STEP PER STEP TECHNICAL GUIDE 

3.1 Enterprise Service Proxy 

Create an Enterprise Service Proxy within the Object Navigator (transaction SE80) by using the corresponding wizard: 

Replace ‘somehost’ with the hostname of the server on which the 3-Heights™Document Converter 

Webservice  is running. 

3.2 Sample Program 

 

*&---------------------------------------------------------------------* 

*& Report  ZDEMO_CONVERT_DOCUMENT 

*& 

*&---------------------------------------------------------------------* 

*& This abap sample shows how to use the Document Converter web service 

*& interface.  

*&   Prerequisites for this program to run: 

*&   For the service consumer proxy ZDEMO_CO_CONVERTER_SOAP a logical  

*&   portname ‘ZDEMO_CONVERTER’ has been defined in SOA Management. 

*&---------------------------------------------------------------------* 

REPORT zdemo_convert_document. 

 

TYPES data_line TYPE x LENGTH 255. 

 

DATA: lt_file_table   TYPE filetable, 

 

*     Filenames ans paths 

      l_file_str_in   TYPE string, 

      l_file_str_out  TYPE string, 

      l_file_path_in  TYPE draw-filep, 

      l_path_in       TYPE draw-filep, 

      l_file_in       TYPE draw-filep, 

      l_file_pure_in  TYPE draw-filep, 

 

*     File data 

      ls_data_line    TYPE data_line, 

      lt_data_tab_in  TYPE STANDARD TABLE OF data_line WITH EMPTY KEY, 

      lt_data_tab_out TYPE STANDARD TABLE OF data_line WITH EMPTY KEY, 

      l_stream_in     TYPE xstring, 

      l_stream_out    TYPE xstring, 

      l_len           TYPE i, 

 

*     Converter 

      lc_conv         TYPE REF TO zdemo_co_converter_soap, 

      ls_soa_in       TYPE zdemo_convert_file_soap_in, 

      ls_soa_out      TYPE zdemo_convert_file_soap_out, 

      l_options       TYPE string, 

      l_final_msg     TYPE string, 

      l_rc            TYPE i. 

* Choose file to convert using dialog 

CALL METHOD cl_gui_frontend_services=>file_open_dialog 

  EXPORTING 

    window_title            = 'Choose file to convert' 

    file_filter             = '*' 

    initial_directory       = 'C:\TEMP' 

    multiselection          = ' ' 

  CHANGING 

    file_table              = lt_file_table 

    rc                      = l_rc 

  EXCEPTIONS 

    file_open_dialog_failed = 1 

    cntl_error              = 2 

    error_no_gui            = 3 

    not_supported_by_gui    = 4 

    OTHERS                  = 5. 

IF sy-subrc = 0. 

  READ TABLE  lt_file_table INTO l_file_str_in INDEX 1. 

ENDIF. 

 

* Load file from client 

IF l_file_str_in IS NOT INITIAL. 

  CALL FUNCTION 'GUI_UPLOAD' 

    EXPORTING 

      filename                = l_file_str_in 

      filetype                = 'BIN' 

    IMPORTING 

      filelength              = l_len 

    TABLES 

      data_tab                = lt_data_tab_in 

    EXCEPTIONS 

      file_open_error         = 1 

      file_read_error         = 2 

      no_batch                = 3 

      gui_refuse_filetransfer = 4 

      invalid_type            = 5 

      no_authority            = 6 

      unknown_error           = 7 

      bad_data_format         = 8 

      header_not_allowed      = 9 

      separator_not_allowed   = 10 

      header_too_long         = 11 

      unknown_dp_error        = 12 

      access_denied           = 13 

      dp_out_of_memory        = 14 

      disk_full               = 15 

      dp_timeout              = 16 

      OTHERS                  = 17. 

* Create instance of the proxy class 

  TRY. 

      CREATE OBJECT lc_conv 

        EXPORTING 

          logical_port_name = 'ZDEMO_CONVERTER'. 

    CATCH cx_ai_system_fault . 

      MESSAGE 'Error: no connection to converter could be established ' TYPE 'I'. 

      RETURN. 

  ENDTRY. 

 

 

* Do some preparations for the service request 

 

  " Build xstring for the given data 

  CLEAR l_stream_in. 

  LOOP AT lt_data_tab_in INTO ls_data_line. 

    CONCATENATE l_stream_in ls_data_line INTO l_stream_in IN BYTE MODE. 

  ENDLOOP. 

  l_stream_in = l_stream_in(l_len). 

 

  " get filename and path of the given file 

  l_file_path_in = l_file_str_in. 

  CALL FUNCTION 'CV120_SPLIT_PATH' 

    EXPORTING 

      pf_path  = l_file_path_in 

    IMPORTING 

      pfx_path = l_path_in 

      pfx_file = l_file_in. 

 

  CALL FUNCTION 'CV120_SPLIT_FILE' 

    EXPORTING 

      pf_file  = l_file_in  " File Name 

    IMPORTING 

      pfx_file = l_file_pure_in. 

 

 

  " Build job options 

  CONCATENATE 'ORIGINALNAME=' l_file_path_in INTO l_options. 

 

  ls_soa_in-doc_bytes = l_stream_in. 

  ls_soa_in-options = l_options. 

* Do conversion 

  TRY. 

      lc_conv->convert_file( 

        EXPORTING 

          input              =  ls_soa_in 

         IMPORTING 

          output             =  ls_soa_out 

      ). 

    CATCH cx_ai_system_fault. 

      MESSAGE 'Error: no connection to converter could be established ' TYPE 'I'. 

      RETURN. 

  ENDTRY. 

 

  IF ls_soa_out-convert_file_result-num_pages GT 0. 

    l_stream_out = ls_soa_out-convert_file_result-document_bytes. 

  ELSE. 

     MESSAGE ls_soa_out-convert_file_result-error_description TYPE 'I'. 

  ENDIF. 

 

 

* Convert output to binary 

  CALL FUNCTION 'SCMS_XSTRING_TO_BINARY' 

    EXPORTING 

      buffer     = l_stream_out 

    TABLES 

      binary_tab = lt_data_tab_out. 

 

 

* Set Output filename 

  CONCATENATE l_path_in l_file_pure_in '.pdf' INTO l_file_str_out. 

 

 

* Download file to client 

  CALL METHOD cl_gui_frontend_services=>gui_download 

    EXPORTING 

      filename = l_file_str_out 

      filetype = 'BIN' 

    CHANGING 

      data_tab = lt_data_tab_out 

    EXCEPTIONS 

      OTHERS   = 1. 

 

* Show result in message dialog 

  CONCATENATE 'Conversion successfull: Check file ' l_file_str_out INTO l_final_msg RESPECTING BL

ANKS. 

  MESSAGE l_final_msg TYPE 'I'. 

 

ENDIF. 


